
TEM Imaging of Annealed AFM Probes
Figure S 1: Example TEM images of two different Si AFM probes, both annealed for 20 min at 1000°C to grow a thick oxide layer.  Oxide thickness was always much thicker (>20nm) than the wear depth of the experiments (5nm). Tip radii varied, likely depending on initial probe apex geometry, but were typically 15-25 nm.  Wear during experiments would typically result in a nominally flat punch geometry.


Novel velocity-dependence measurement technique

Rapid AFM tip wear during experiments necessitated the use of an alternative technique for measuring the velocity dependence of kinetic friction, as discussed in the main text, which could capture the velocity dependence of friction with a minimum of sliding.  The solution was to span the entire velocity range in one sliding cycle.  An exponential waveform, which most commercial waveform generators can output without modification, was deemed unsuitable because the vast majority of the sliding track would be traversed at the highest sliding velocities.  Instead, we wanted each segment of the track to cover an equal increment of the velocity range of interest. Solving the following differential equation will deliver such a waveform:

where  is the track position,  is time, and , ,  are constants to determine the starting position and track length to reach the desired maximum velocity. This allows for the track velocity to increase exponentially as a function of track length traversed, not elapsed time. This equation can be solved analytically using Mathematica (V12.1, Wolfram, Champaign, IL) to yield:

This transcendental solution is difficult to apply, so we chose to numerically solve the differential equation in IgorPro to generate the ¼ cycle waveform.Figure S 2: Example ¼ cycle waveform

This waveform could then be transformed several ways to form the four segments of a full cycle. One possible example is shown in Fig. S3Figure S 3: Example full scan waveform for speed-dependent friction measurement.

However, the transformations were chosen so that the following goals were achieved: (1) the cycle starts at scanner zero so there are no instantaneous jumps in position when starting the test; (2) every track position experiences the same tip velocity for either direction of tip motion, allowing for the standard friction calculation from one half of the friction force loop width; (3) the slowest velocities occur at the tip direction reversal points to prevent the excitation of scanner resonances which add noise to the force trace; and (4) each tip velocity occurs two times per cycle, so each velocity is sampled at two different track positions. The resulting velocity dependence is as follows, determined by differentiating the scanning waveform:
Figure S 4: Tip speed as a function of track position during test.  The left panel uses a log speed axis and shows the positive speeds (the first half of the cycle to show that each exponentially-spaced speed increment is sampled from an equal track length.  The right panel shows the same data covering the entire sliding cycle but with a linear axis for speed..

	To run the test, the scanner excitation signal, after converting to the appropriate voltage given the piezo sensitivity (determined via a separate calibration),[1] was output directly from IgorPro using the NIDAQToolsMX Igor module which can communicate with National Instruments hardware.  The signal was sent to an NI USB-6259 (National Instruments, Austin, TX), which subsequently output an analog signal via USB to the R9 AFM controller, where the analog signal was amplified and sent to the X-piezo. The raw analog lateral force signal was sent via BNC from the R9, through the NI USB-6259, and captured in the Igor software.
	Each test consisted of 8 cycles of reciprocating motion that were run in a single uninterrupted series. Since the NIDAQToolsMX allows for triggered data capture, the data trace could be correlated directly to a corresponding track position and tip velocity in the scanning waveform. This allowed for automatic analysis of the data, which is essentially a necessity, given the complexity of the data structure.  The codes to generate the scan waveform, to perform the test, and to perform the analysis are provided below.  All code is Igor-specific but could be readily modified to perform the same tasks using e.g. Matlab.

Generating the scan wave

Function FillWave(TestWave,del_t,y0,v0,v2)

Wave TestWave
Variable del_t,y0,v0,v2
Variable n

for (n=1; n<dimsize(TestWave,0); n+=1)
	TestWave[n][0] = TestWave[n-1][0] + ( y0 + (v0-y0)*((v2-y0)/(v0-y0))^(TestWave[n-1][0]/1000))*del_t
			//Testwave set for 2000nm
	if (TestWave[n][0] >= 1000)  //Stop when tip reaches 500nm
		break
	endif
endfor

end

The subsequent manipulation of the ¼ cycle to generate a full cycle was performed from the command line, but can be accomplished as follows, for the case of a 2 µm track length, with a 10 nm/s to 100 μm/s waveform:

Make/N=3e7 QtrTrack1000nm10nms100ums
SetScale/P x 0,1e-6,"", QtrCycle1000nm10nms100ums
FillWave(QtrCycle1000nm10nms100ums,1e-6,0.1,10,100000)
	//not that the FillWave procedure needs to be modified slightly from 
	//above for this track length
DeletePoints 10934807,29999999, QtrCycle1000nm10nms100ums
	//Need to delete points not used by FillWave()
Duplicate QtrCycle1000nm10nms100ums_4thQtr QtrCycle1000nm10nms100ums_1stQtr,QtrCycle1000nm10nms100ums_2ndQtr,QtrCycle1000nm10nms100ums_3rdQtr
QtrCycle1000nm10nms100ums_4thQtr = QtrCycle1000nm10nms100ums_4thQtr - 1000.00952148438
FillWaveBackwards(QtrCycle1000nm10nms100ums_4thQtr,QtrCycle1000nm10nms100ums_3rdQtr)
QtrCycle1000nm10nms100ums_1stQtr = - QtrCycle1000nm10nms100ums_3rdQtr
QtrCycle1000nm10nms100ums_2ndQtr = - QtrCycle1000nm10nms100ums_4thQtr
Concatenate/NP {QtrCycle1000nm10nms100ums_1stQtr, QtrCycle1000nm10nms100ums_2ndQtr, QtrCycle1000nm10nms100ums_3rdQtr,QtrCycle1000nm10nms100ums_4thQtr}, FullCycle2000nm10nms100ums_nm
Differentiate FullCycle2000nm10nms100ums_nm/D=FullCycle2000nm10nms100ums_nm_DIF
	// This wave is needed for the auto analysis
Duplicate FullCycle2000nm10nms100ums_nm FullCycle2000nm10nms100ums_V
	//to convert to from nm to volts for the electronics
FullCycle2000nm10nms100ums_V = FullCycle2000nm10nms100ums_V / 23.5191
	//23.5191 is the nm to V conversion known via lateral calibration
FullCycle2000nm10nms100ums_V = FullCycle2000nm10nms100ums_V / 15
	// the R9 has a 15x amplifier

One of the commands above makes use of a trivial auxiliary procedure:

Function FillWaveBackwards(SourceWave,DestWave)

Wave SourceWave,DestWave
Variable n
for (n=1; n<numpnts(SourceWave)+1; n+=1)
	DestWave[n] =  SourceWave[numpnts(SourceWave)-1-n]
endfor
end

Running the Test

This task/code requires use of NIDAQToolsMX.

Function RunSpeedDepTest(WGwave)
Wave WGwave
String WGwavePath = GetWavesDataFolder(WGwave, 2)
//String WGwavePath
String WGParamString = WGwavePath+",0"

Make/N=(numpnts(WGwave)*deltax(WGwave)*8*1000+2000) LF		//8 cycles in these test, a few extra points in case synchronization is not perfect
SetScale/P x, 0,0.001, "s", LF		//scales data for 1kHz sampling in test


//Now start actual test and acquisition, hopefully this part happens quickly
	//Set up data acq
DAQmx_Scan /DEV="dev1"/AVE=100/BKG/TRIG={"/Dev1/ao/StartTrigger"} WAVES="LF, 0/DIFF,-0.5,0.5"
	//set up waveform generation, see NIDAQToolsMX documentation for details
DAQmx_WaveformGen /DEV="dev1" /NPRD=8  WGParamString

end

Analyzing the Data

This analysis is specific to the 1 µm track length, 10 nm/s to 10 μm/s waveform, but can be easily modified to accommodate other velocity ranges and track lengths.

Function CalculateFriction(LFWave,SpeedWave,PositionWave,PeriodPts,NumPds)
Wave LFWave, SpeedWave, PositionWave
Variable PeriodPts, NumPds	//can figure these out from the scanwave
Variable m,n,a,b

	// Calculate the basic friction point-by-point, cycle-by-cycle
Make/O/N=(PeriodPts/2,NumPds) FrictionCalc
for (m=0; m<NumPds; m+=1)
	for (n=0; n<PeriodPts/4-1; n+=1)
		FrictionCalc[n][m] = abs(LFWave[n+m*PeriodPts]-LFWave[PeriodPts/2-1-n+m*PeriodPts])/2
		FrictionCalc[n+PeriodPts/4][m] = abs(LFWave[n+PeriodPts/2+m*PeriodPts]-LFWave[PeriodPts-1-n+m*PeriodPts])/2
	endFor
endFor

	//Calculate average of two spots with same tip speed, cycle-by-cycle
Make/O/N=(round(PeriodPts/4)-1,NumPds) FrictionCalc_Avg1
for (m=0; m<NumPds; m+=1)
	for (n=0; n<round(PeriodPts/4)-1; n+=1)
		FrictionCalc_Avg1[n][m] = (FrictionCalc[n][m]+FrictionCalc[n+round(PeriodPts/4)-1][m])/2
	endFor
endFor

	//Calculate average tip speed + STD for every 10nm segment of track
a=0
Make/O/N=(50,2) TipSpeed_Sliced
for (n=0; n<50; n+=1)
	FindLevel/P/Q PositionWave, 10*n+10		//get point range for speed track segment average
	b = V_LevelX								// Set top of track segment range in points
	Wavestats/Q/P/R=[a,b-1] SpeedWave	//get average+RMS speed for segment
	TipSpeed_Sliced[n][0] = V_avg			//store values
	TipSpeed_Sliced[n][1] = V_sdev
	a=b
endFor	
	
	
	//Calculate average friction + STD for every 10nm segment of track, cycle-by-cycle
a=0
Make/O/N=(50,NumPds) FrictionCalc_Sliced, FrictionCalc_Sliced_STD
for (m=0; m<NumPds; m+=1)
	for (n=0; n<50; n+=1)
		FindLevel/P/Q PositionWave, 10*n+10						//get point range for speed track segment average
		b = V_LevelX												// Set top of track segment range in points
		Wavestats/Q/P/RMD=[a,b-1][m] FrictionCalc_Avg1	//get average+RMS friction for segment
		FrictionCalc_Sliced[n][m] = V_avg					//store data
		FrictionCalc_Sliced_STD[n][m] = V_sdev
	endFor
endFor
end

The cycle-by-cycle data is averaged as desired using the following code:

Function AverageCycleFricsV2(PositionWave, FrictionCalc_Avg1,CycleStart,CycleEnd)
Wave PositionWave, FrictionCalc_Avg1
Variable CycleStart,CycleEnd
Variable n,a, b

Make/O/N=(50) FrictionCalc_Sliced_AVG, FrictionCalc_Sliced_AVG_STD

a=0
for (n=0; n<50; n+=1)
	FindLevel/P/Q PositionWave, 10*n+10						//get point range for speed track segment average
	b = V_LevelX												// Set top of track segment range in points
	Wavestats/Q/P/RMD=[a,b-1][CycleStart-1,CycleEnd-1] FrictionCalc_Avg1	//get average+RMS friction for segment
	FrictionCalc_Sliced_AVG[n] = V_avg					//store data
	FrictionCalc_Sliced_AVG_STD[n] = V_sdev
endFor
end



Benchmarking the Velocity Dependence Measurement

To confirm that the new technique provides equivalent results to the typical velocity dependence measurement technique where images are captured at a selection of distinct sliding speeds so the average friction at each speed can be extracted, both techniques were utilized sequentially on the same HOPG sample (Structure Probe, Inc., West Chester, PA) with a DLC-coated AFM probe (contDLC, Budget Sensors, Sofia, Bulgaria). Kinetic friction is known to scale with the logarithm of the sliding velocity across the velocity range examined here on HOPG.[2] The new protocol was tested for a 1 μm track length from 10 nm/s to 10 μm/s, and also for a 2 μm track length from 10 nm/s to 100 μm/s. The typical protocol was performed at a selection of discreet velocities by reciprocating scanning across a 100, 500, or 2000 nm track length depending on the scan velocity, and by averaging the friction across at least 4 sliding cycles. As seen is Fig. S5, the results are equivalent. The perturbation in the trendline for the larger velocity range trace was due to the probe traversing grain boundaries on the sample.
Figure S 5: Comparison of speed dependence of kinetic friction on HOPG for the new protocol discussed above and the standard protocol of capturing images at a variety of discreet sliding speeds.

























Cycle-by-cycle evolution at low % RH

The main text discusses an evolution effect in the velocity-dependent friction at low % RH (shown in Fig. 6). Figure S 6 below shows another example where friction evolved to extinguish velocity weakening behavior across a test, in this case at low chamber total pressure, as compared to a test at 49±2% RH where changes during the test were much less pronounced and the qualitative shape of the velocity dependence did not change.[bookmark: _Ref139288499]Figure S 6: a) All cycles of a speed-dependent friction test performed at 1 mbar total pressure i.e. <<1% RH. An obvious evolution in friction response occurs during the test. b) All cycles of a speed dependent friction test performed at 49% RH. 




Tip Geometry after Pressure-Dependence Tests

After performing the pressure dependence testing shown in Fig. 4 of the main text, AFM topographic imaging of an ultrananocrystalline diamond (UNCD) sample was performed in tapping mode. Tapping mode was chosen as it tends to yield better topographic fidelity due to the smaller real tip-surface contact area, relative to contact mode.[3]  The very small grain size of UNCD, 3-5 nm,[4] relative to the large radius of these oxidized tips after wear during experiments, makes the error introduced by tip-sample convolution small.[5, 6] To determine the tip radius, 3 linescans across the apexes of small UNCD grains were extracted and a circle function to the top 1 nm of the apex to find the radius of the probe: 87.7±4.5 nm.  This radius was used for the DMT calculations discussed in the main text.




Figure S 7: (a) AFM topographic image of UNCD captured in tapping mode immediately after capturing the data presented in Fig. 7 of the manuscript. (b) Fitting of the apex with dashed lines in linescan 1 shown in (a), (blue trace), along with an equivalent measurement made immediately before the high load experiment (red trace).

